

*********************************************************************************************************
*** Replication data for: Rebel Recruitment and Migration - Theory and Evidence from Southern Senegal ***
*** Publication: Journal of Conflict Resolution                                                       ***
*** Authors: Max Schaub and Daniel Auer                                                               ***
*********************************************************************************************************


// Instructions and setup

* 1. Put the datasets (rebelmig.dta, rebelmig_ff.dta, and rebelmig_gambia.dta in a folder
* 2. Set your working directory to this folder
cd  "..."
* 3. Execute this do-file

* Install packages 
ssc install coefplot, replace // thanks to Ben Jann
ssc install estout, replace // thanks to Ben Jann (once more)
ssc install blindschemes, replace // thanks to Daniel Bischof


* Graphics setup
set scheme s1mono

* Create folder for figures and tables
mkdir Figures
mkdir Tables


* Figure 1

use "rebelmig.dta", clear
graph bar, over(recruitconcern)  title("Concerned about rebel recruitment", size(medlarge) pos(12)) ytitle("Percent") blabel(bar, format(%9.0fc) size(vsmall)) horizontal bargap(1) name(recruitconcern, replace) fysize(40)
graph bar, over(rebelconcern)  title("Concerned about rebel activities", size(medlarge) pos(12)) ytitle("Percent") blabel(bar, format(%9.0fc) size(vsmall)) horizontal bargap(1) name(rebelconcern, replace) fysize(40)
graph bar, over(rebelsuffer)  title("Suffered from violence", size(medlarge) pos(12)) ytitle("Percent") blabel(bar, format(%9.0fc) size(vsmall)) horizontal bargap(50) name(rebelsuffer, replace) fysize(30)
graph combine rebelconcern recruitconcern rebelsuffer, col(1) xcommon name(rebeldirect, replace) 
graph export "Figures/fig1.pdf", replace


* Figure 2

use "rebelmig.dta", clear
graph bar migplan, over(recruitconcern, label(labsize(small))) ytitle("Migration plans respondent", size(large)) ylab(, nogrid format(%3.2f)) b1title("Concern about rebel recruitment", size(medlarge) margin(medlarge)) name(recruitplan, replace)
graph bar migsupportself, over(recruitconcern, label(labsize(small))) ytitle("Family support for migration", size(large)) ylab(, nogrid format(%3.2f))  b1title("Concern about rebel recruitment", size(medlarge) margin(medlarge)) name(recruitsupport, replace) 
graph combine recruitplan recruitsupport, row(1)
graph display, xsize(2.7) ysize(1) scale(1.2) scheme(s1mono) 
graph export "Figures/fig2.pdf", replace


* Figure 4d

use "rebelmig.dta", clear
twoway (scatter distbase tmc10km) (lfit distbase tmc10km, ytitle("Distance to nearest MFDC base according to ACLED") ylab(, format(%9.0fc) nogrid) xlab(, nogrid) note("") title("") legend(off) ), graphregion( margin(vsmall)) scheme(plotplain)
graph export "Figures/fig4d.pdf", replace


* Table 1

use "rebelmig.dta", clear
eststo main1: reg migplan distbase10 female age edufather edumother ib3.ethnicity, r
eststo main2: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity, r
eststo main3: reg migplan distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network, r
eststo main4: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network, r
eststo main5: reg migplan distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_97_100, r
eststo main6: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_97_100, r
eststo main7: reg migplan distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_97_100 i.strata, r
eststo main8: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_97_100 i.strata, r

esttab main1 main3 main5 main7 main2 main4 main6 main8 using Tables/tab1.rtf, ///
order(distbase10) ///
compress se label star(`=uchar(8224)' 0.1 * 0.05 ** 0.01) b(%9.2f) sfmt(%9.2gc) wrap varwidth(25) r2 ///
nomtitle  stats(N r2, fmt(%9.0gc %9.2f)) nogaps ///
indicate("Ethnic group FE = *ethnicity" "Strata FE = *strata" , labels(yes "--")) 


* Table 2

use "rebelmig.dta", clear
eststo iv0: reg distbase10 tmc10km female age edufather edumother ib3.ethnicity mfdcfat_97_100, r
eststo iv1: reg migplan tmc10km female age edufather edumother ib3.ethnicity mfdcfat_97_100, r
eststo iv2: ivregress 2sls migplan female age edufather edumother ib3.ethnicity mfdcfat_97_100 (distbase10=tmc10km), r
estat firststage
eststo iv3: reg migsupportself tmc10km female age edufather edumother ib3.ethnicity mfdcfat_97_100, r
eststo iv4: ivregress 2sls migsupportself female age edufather edumother ib3.ethnicity mfdcfat_97_100 (distbase10=tmc10km), r

esttab iv0 iv1 iv2 iv3 iv4  using Tables/tab2.rtf, ///
order(tmc10km distbase10) ///
compress se label star(`=uchar(8224)' 0.1 * 0.05 ** 0.01) b(%9.2f) sfmt(%9.2gc) wrap varwidth(25) r2 ///
nomtitle  stats(N r2, fmt(%9.0gc %9.2f)) nogaps   ///
indicate("Ethnic group FE = *ethnicity", labels(yes "--")) 


* Figure 5

use "rebelmig_ff.dta", clear
* Note: Allow some time for these models to run

* overall
gsem (i.migstat <- c.tmc10km c.mfdcfat_97_100 i.femalerelative i.agecat M1[parent_key]@1), mlogit vce(cluster parent_key)
eststo Domestic: margins, dydx(tmc10km) post 
gsem (i.migstatalt <- c.tmc10km c.mfdcfat_97_100 i.femalerelative i.agecat M1[parent_key]@1), mlogit vce(cluster parent_key)
eststo International: margins, dydx(tmc10km)  post coeflegend

coefplot Domestic, keep(*:2._predict*) bylabel(Domestic) || ///
		 (International, keep(*:2._predict*)), ///
		  ytitle("Overall") xline(0) plotr(m(medlarge)) ///
		  ylabel(1 "Placehol", tlength(0) labcolor(white)) ///
		  xscale(range(-0.1 (0.1) 0.2)) xlabel(-0.1 (0.1) 0.22, format(%04.2f) nogrid) grid(none) ///
		  subtitle(, size(large)  justification(center) ///
		   color(black) bcolor(white) bmargin(bottom) lcolor(white)) ///
		  msymbol(D) msize(medsmall) mfcolor(white) mlcolor(gs7) ///
		  name(tmcoverall, replace) fysize(20)
* by gender
gsem (i.migstat <- c.tmc10km##i.femalerelative c.mfdcfat_97_100 i.agecat M1[parent_key]@1), mlogit vce(cluster parent_key)
eststo genderdom: margins, dydx(tmc10km) at(femalerelative==(0 1)) post coeflegend
gsem (i.migstatalt <- c.tmc10km##i.femalerelative c.mfdcfat_97_100 i.agecat M1[parent_key]@1), mlogit vce(cluster parent_key)
eststo genderint: margins, dydx(tmc10km) at(femalerelative==(0 1))  post coeflegend

cefplot (genderdom, keep(*:2._predict*)) || ///
		 (genderint, keep(*:2._predict*) ), ///
		  coeflabels(2._predict#1._at = "Male" 2._predict#2._at = "Female") ///
		  ytitle("By gender") xline(0) plotr(m(medlarge)) ///
		  xscale(range(-0.1 (0.1) 0.2)) xlabel(-0.1 (0.1) 0.22, format(%04.2f) nogrid) grid(none) ///
		  subtitle("", size(medlarge)  justification(center) ///
		   color(black) bcolor(white) bmargin(bottom) lcolor(gs7)) ///
		  msymbol(D) msize(medsmall) mfcolor(white) mlcolor(gs7) ///
		  name(tmcgender, replace) fysize(30)
* by age group
gsem (i.migstat <- c.tmc10km##i.agecat c.mfdcfat_97_100 i.femalerelative M1[parent_key]@1), mlogit vce(cluster parent_key)
eststo agedom: margins, dydx(tmc10km) at(agecat==(0 1 2 3 4 5 6))  post coeflegend
gsem (i.migstatalt <- c.tmc10km##i.agecat c.mfdcfat_97_100 i.femalerelative M1[parent_key]@1), mlogit vce(cluster parent_key)
eststo ageint: margins, dydx(tmc10km) at(agecat==(0 1 2 3 4 5 6))  post coeflegend

coefplot agedom, keep(*:2._predict*) bylabel(Domestic) || ///
		 (ageint, keep(*:2._predict*) label("15-20")), ///
		 drop(2._predict#1._at) /// //leave out the youngest category; technically, this shouldn't exist since only family members aged 15 and above should be recorded
		  coeflabels(2._predict#1._at = "Under 15" 2._predict#2._at = "15 to 19" 2._predict#3._at =  "20 to 24" ///
			2._predict#4._at = "25 to 29" 2._predict#5._at = "30 to 39" 2._predict#6._at = "40 to 49" 2._predict#7._at = "Over 49") ///
		  ytitle("By age group") xline(0) plotr(m(medlarge)) ///
		  xscale(range(-0.1 (0.1) 0.2)) xlabel(-0.1 (0.1) 0.22, format(%04.2f) nogrid) grid(none) ///
		  subtitle("", size(medlarge)  justification(center) ///
		   color(black) bcolor(white) bmargin(bottom) lcolor(gs7)) ///
		  msymbol(D) msize(medsmall) mfcolor(white) mlcolor(gs7) ///
		  name(tmcage, replace) fysize(50)
	
graph combine tmcoverall tmcgender tmcage, col(1) row(3) name(tmcall, replace)
graph export "Figures/fig5.pdf", replace


* Figure 6 

use "rebelmig.dta", clear
eststo hetreldm: reg migsupportself c.tmc10km##c.stdeconomy age edufather edumother ib3.ethnicity  mfdcfat_97_100  if female==0, coefl r
margins, dydx(tmc10km) at(stdeconomy==(-1 (.5) 1.5))
marginsplot, recast(line) ciopt(color(%20)) recastci(rarea) yline(0) ytitle("") xtitle("") ylabel(-0.4 (0.4) 0.8, format(%02.1f)) xlabel(, format(%02.1f)) title("Deprivation", size(medium)) name(deprivation, replace) 
eststo hetmargm: reg migsupportself c.tmc10km##c.stdrejectsenegal age edufather edumother ib3.ethnicity  mfdcfat_97_100  if female==0, coefl r
margins, dydx(tmc10km) at(stdrejectsenegal==(-.5 (.5) 2.5))
marginsplot, recast(line) ciopt(color(%20)) recastci(rarea) yline(0) ytitle("") xtitle("") ylabel(, format(%02.1f)) xlabel(, format(%02.1f)) title("Alienation", size(medium)) name(alienation, replace)
eststo hetlandm: reg migsupportself c.tmc10km##c.stdnolandowner age edufather edumother ib3.ethnicity  mfdcfat_97_100  if female==0, coefl r
margins, dydx(tmc10km) at(stdnolandowner==(-1 (.5) 1))
marginsplot, recast(line) ciopt(color(%20)) recastci(rarea) yline(0) ytitle("",  margin(l+5.5)) xtitle("") ylabel(, format(%02.1f)) xlabel(, format(%02.1f)) title("Poverty", size(medium)) name(poverty, replace) 
eststo hetsocialm: reg migsupportself c.tmc10km##c.stdsocialnet female age edufather edumother ib3.ethnicity  mfdcfat_97_100  if female ==0, coefl r
margins, dydx(tmc10km) at(stdsocialnet==(-1 (.5) 2.5))
marginsplot, recast(line) ciopt(color(%20)) recastci(rarea) yline(0) ytitle("") xtitle("") ylabel(, format(%02.1f)) xlabel(, format(%02.1f)) title("Social embeddedness", size(medium)) name(social, replace)

graph combine deprivation alienation poverty social, col(2) name(hetmigplanm, replace) title("Family support for migration, males", size(medsmall))

eststo clear
eststo hetreldf: reg migsupportself c.tmc10km##c.stdeconomy age edufather edumother ib3.ethnicity mfdcfat_97_100  if female==1, coefl r
margins, dydx(tmc) at(stdeconomy==(-1 (.5) 1.5))
marginsplot, recast(line) ciopt(color(%20)) recastci(rarea) yline(0) ytitle("") xtitle("") ylabel(, format(%02.1f)) xlabel(, format(%02.1f))  title("Deprivation", size(medium)) name(deprivation, replace)
eststo hetmargf: reg migsupportself c.tmc10km##c.stdrejectsenegal age edufather edumother ib3.ethnicity mfdcfat_97_100  if female==0, coefl r
margins, dydx(tmc) at(stdrejectsenegal==(-.5 (.5) 2.5))
marginsplot, recast(line) ciopt(color(%20)) recastci(rarea) yline(0) ytitle("") xtitle("") ylabel(-0.3 (0.3) 0.6, format(%02.1f)) xlabel(, format(%02.1f))  title("Alienation", size(medium)) name(alienation, replace)
eststo hetlandf: reg migsupportself c.tmc10km##c.stdnolandowner age edufather edumother ib3.ethnicity mfdcfat_97_100  if female==1, coefl r
margins, dydx(tmc) at(stdnolandowner==(-1 (.5) 1))
marginsplot, recast(line) ciopt(color(%20)) recastci(rarea) yline(0) ytitle("") xtitle("") ylabel(, format(%02.1f)) xlabel(, format(%02.1f))  title("Poverty", size(medium)) name(poverty, replace)
eststo hetsocialf: reg migsupportself c.tmc10km##c.stdsocialnet female age edufather edumother ib3.ethnicity mfdcfat_97_100  if female ==1, coefl r
margins, dydx(tmc) at(stdsocialnet==(-1 (.5) 2.5))
marginsplot, recast(line) ciopt(color(%20)) recastci(rarea) yline(0) ytitle("") xtitle("") ylabel(-0.4 (0.4) 0.15, format(%02.1f)) xlabel(, format(%02.1f))  title("Social embeddedness", size(medium)) name(social, replace)

graph combine deprivation alienation poverty social, col(2) name(hetmigplanf, replace) title("Family support for migration, females", size(medsmall))

graph combine hetmigplanm hetmigplanf, col(1) name(het, replace) note("{it:Note:} Marginal effect of canopy cover given different values of the depicted, standardized indicators.", size(tiny))
graph display het, ysize(1.3) xsize(1) scale(.9) scheme(s1mono) 
graph export "Figures/fig6.pdf", replace



// Appendix

* Table 3 

use "rebelmig.dta", clear
tab ethnicity, gen(ethn)
label variable ethnicity "Ethnicity"
label variable unempl "Respondent unemployed"
label variable hhecon "Household's economic situation"
label variable network "Nr of network contacts abroad"
label variable distbase10 "Distance to destroyed MFDC base in 10km"
label variable economy "Assessment of state of economy"
label variable northerner "Northerner"
local varsind female age edufather edumother ethn3 ethn4  ethn2 ethn11 ethn7 ethn1 ethn6 ethn5  ethn8 ethn9 ethn10 unempl hhecon network migplan migsupportself rebelconcern recruitconcern rebelsuffer economy rejectsenegal landowner socialnet distbase10 tmc10km mfdcfat_97_100 
estpost summarize `varsind'
esttab .  using Tables/tab3.rtf, cells("mean(fmt(2)) sd(fmt(2)) min(fmt(2)) max(fmt(2)) count(fmt(0))")  nonumber varwidth(45) wrap replace label noobs 


* Table 4

use "rebelmig_ff.dta", clear
tab agecat, gen(age)
tab migstat, gen(mig)
label variable femalerelative "Female family member"
local varsfam femalerelative age1 age2 age3 age4 age5 age6 age7 mig1 mig2 mig3 tmc10km mfdcfat_97_100
estpost summarize `varsfam'
esttab .  using Tables/tab4.rtf, cells("mean(fmt(2)) sd(fmt(2)) min(fmt(2)) max(fmt(2)) count(fmt(0))")  nonumber varwidth(45) wrap replace label noobs 


* Table 5

use "rebelmig.dta", clear
eststo concernplan: reg migplan rebelconcern female age edufather edumother, r
eststo concernsupport: reg migsupportself rebelconcern female age edufather edumother, r
eststo recruitplan: reg migplan recruitconcern female age edufather edumother, r
eststo recruitsupport: reg migsupportself recruitconcern female age edufather edumother, r
eststo sufferplan: reg migplan rebelsuffer female age edufather edumother, r
eststo suffersupport: reg migsupportself rebelsuffer female age edufather edumother, r

esttab concernplan recruitplan sufferplan concernplan recruitsupport sufferplan  using Tables/tab5.rtf, ///
order(rebelconcern recruitconcern rebelsuffer) ///
compress se label star(`=uchar(8224)' 0.1 * 0.05 ** 0.01) b(%9.2f) sfmt(%9.2gc) wrap varwidth(25) r2 ///
nomtitle  stats(N r2, fmt(%9.0gc %9.2f)) nogaps   


* Table 6

use "rebelmig.dta", clear
eststo conalt1: reg migplan distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network distgov, r
eststo conalt2: reg migplan distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_10_100, r
eststo conalt3: reg migplan distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_15_100, r
eststo conalt4: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network distgov, r 
eststo conalt5: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_10_100, r
eststo conalt6: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_15_100, r

esttab conalt1 conalt2 conalt3 conalt4 conalt5 conalt6  using Tables/tab6.rtf, ///
order(distbase10) ///
compress se label star(`=uchar(8224)' 0.1 * 0.05 ** 0.01) b(%9.2f) sfmt(%9.2gc) wrap varwidth(25) r2 ///
nomtitle  stats(N r2, fmt(%9.0gc %9.2f)) nogaps ///
indicate("Ethnic group FE = *ethnicity" , labels(yes "--")) 


* Table 7

use "rebelmig.dta", clear
eststo nmain1: reg migplan distbase10 female age edufather edumother ib3.ethnicity if northerner==3, r
eststo nmain2: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity if northerner==3, r 
eststo nmain3: reg migplan distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network if northerner==3, r
eststo nmain4: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network if northerner==3, r 
eststo nmain5: reg migplan distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_97_100 if northerner==3, r 
eststo nmain6: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_97_100 if northerner==3, r 
eststo nmain7: reg migplan distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_97_100 i.strata if northerner==3, r
eststo nmain8: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity unempl hhecon network mfdcfat_97_100 i.strata if northerner==3, r

esttab nmain1 nmain3 nmain5 nmain7 nmain2 nmain4 nmain6 nmain8  using Tables/tab7.rtf, ///
order(distbase10) ///
compress se label star(`=uchar(8224)' 0.1 * 0.05 ** 0.01) b(%9.2f) sfmt(%9.2gc) wrap varwidth(25) r2 ///
nomtitle  stats(N r2, fmt(%9.0gc %9.2f)) nogaps replace  ///
indicate("Ethnic group FE = *ethnicity" "Strata FE = *strata" , labels(yes "--")) 


* Table 8

use "rebelmig_gambia.dta", clear
eststo gmain1: reg migplan distbase10 female age edufather edumother ib3.ethnicity, r
eststo gmain2: reg migsupportself distbase10 female age edufather edumother ib3.ethnicity, r
eststo giv1: reg migplan tmc10km female age edufather edumother ib3.ethnicity, r
eststo giv2: reg migsupportself tmc10km female age edufather edumother ib3.ethnicity, r
eststo giv3: ivregress 2sls migplan female age edufather edumother ib3.ethnicity (distbase10=tmc10km), r 
eststo giv4: ivregress 2sls migsupportself female age edufather edumother ib3.ethnicity (distbase10=tmc10km), r 
* R-square for giv4
reg distbase10 female age edufather edumother tmc10km ib3.ethnicity
predict dbhat, xb
reg migsupportself dbhat female age edufather edumother ib3.ethnicity

esttab gmain1 giv1 giv3 gmain2 giv2 giv4  using Tables/tab8.rtf, ///
order(distbase10 tmc10km) ///
compress se label star(`=uchar(8224)' 0.1 * 0.05 ** 0.01) b(%9.2f) sfmt(%9.2gc) wrap varwidth(25) r2 ///
nomtitle  stats(N r2, fmt(%9.0gc %9.2f)) nogaps   ///
indicate("Ethnic group FE = *ethnicity", labels(yes "--")) 


* Figure 7

use "rebelmig_ff.dta", clear
mixed returnmig c.tmc10km c.mfdcfat_97_100 i.femalerelative i.agecat ||parent_key: // simpler, but linear
eststo returnoverall: margins, dydx(tmc10km)  post coeflegend

coefplot returnoverall, keep(*tmc10km*) ///
		  ytitle("Overall") xline(0) plotr(m(medlarge))  ///
		  ylabel(1 "Placehol", tlength(0) labcolor(white)) ///
		  xscale(range(-0.1 (0.1) 0.1)) xlabel(-0.1 (0.1) 0.1, format(%04.2f) nogrid) grid(none) ///
		  subtitle(, size(large)  justification(center) ///
		   color(black) bcolor(white) bmargin(bottom) lcolor(white)) ///
		  msymbol(D) msize(medsmall) mfcolor(white) mlcolor(gs7) ///
		  name(returnoverall, replace) fysize(20)

mixed returnmig c.tmc10km##i.femalerelative c.mfdcfat_97_100 i.agecat ||parent_key:
eststo returngender: margins, dydx(tmc10km) at(femalerelative==(0 1))  post coefl
test _b[tmc10km:1bn._at] = _b[tmc10km:2._at]

coefplot returngender, ///
		  coeflabels(1._at = "Male" 2._at = "Female") ///
		  ytitle("By gender") xline(0) plotr(m(medlarge)) ///
		  xscale(range(-0.1 (0.1) 0.1)) xlabel(-0.1 (0.1) 0.1, format(%04.2f) nogrid) grid(none) ///
		  subtitle("", size(medlarge)  justification(center) ///
		   color(black) bcolor(white) bmargin(bottom) lcolor(gs7)) ///
		  msymbol(D) msize(medsmall) mfcolor(white) mlcolor(gs7) ///
		  name(returngender, replace) fysize(30)

mixed returnmig c.tmc10km##i.agecat c.mfdcfat_97_100 i.femalerelative ||parent_key: if femalerelative==1
eststo returnage: margins, dydx(tmc10km) at(agecat==(0 1 2 3 4 5 6))  post coeflegend

coefplot returnage, ///
		 drop(1._at) /// //leave out the youngest category; technically, this shouldn't exist since only family members aged 15 and above should be recorded
		  coeflabels(1._at = "Under 15"  2._at = "15 to 19" 3._at = "20 to 24" 4._at = "25 to 29" 5._at = "30 to 39" 6._at = "40 to 49" 7._at = "Over 49") ///
		  ytitle("By age group") xline(0) plotr(m(medlarge)) ///
		  xscale(range(-0.1 (0.1) 0.1)) xlabel(-0.1 (0.1) 0.1, format(%04.2f) nogrid) grid(none) ///
		  subtitle("", size(medlarge)  justification(center) ///
		   color(black) bcolor(white) bmargin(bottom) lcolor(gs7)) ///
		  msymbol(D) msize(medsmall) mfcolor(white) mlcolor(gs7) ///
		  name(returnage, replace) fysize(50)
	
graph combine returnoverall returngender returnage, col(1) row(3) name(returnall, replace)
graph export "Figures/fig7.pdf", replace


* Figure 8

use "rebelmig.dta", clear
twoway (lowess recruitconcern distbase, ytitle("Concern with rebel recruitment") xtitle("Distance to former MFDC base (in km)") ylab(, format(%9.0fc) nogrid) xlab(, format(%9.0fc) nogrid) lstyle(width(medthick)) note("") title("") legend(off) ), graphregion( margin(medium))
graph export "Figures/fig8.pdf", replace
 
		
